
Welcome 

Thank you for downloading the third issue of our newsletter. These are difficult 

times for all of us and we hope you, your family, and colleagues are all staying 

safe and well. It is times such as this that we all need to come together to 

support one another, and we are confident that the scientific community will 

adapt to these changes and continue to thrive. 

Be assured that CED is still open for business, and we are continuing our daily 

operations. Our software team is still developing and providing updates for 

Spike2 and Signal, and our hardware team is still at hand for any help you need. 

Our customer support teams are available to answer any technical questions and 

provide remote training via Skype, Zoom or TeamViewer. We may be contacted 

using all the usual channels; however, we ask that contact be done via email 

where possible. If you do need to speak to someone directly, please send us an 

email to request a call back in the first instance. Up to date information can 

always be found on our website, and the CED user forums are also available for 

technical help from our super users. 

We are also endeavouring to add more video tutorials to our existing library and 

would like your input on topics you think would be useful to cover. Please get in 

touch at Marjorie@ced.co.uk and let us know your ideas! 

 

Training Days 

Due to the ongoing COVID-19 outbreaks, all current training events have been 

put on hold. We do however offer remote training sessions via Skype either one-

to-one or with groups. 

Join us and learn how to make the best use of Spike2 and Signal to save hours of 

repetitive analysis. Our remote sessions are free to arrange and are suitable for 

both existing and prospective users of our data acquisition and analysis systems. 

If you would like to schedule a session, please get in touch. 

If you are interested in hosting a training event in your local area once social 

distancing measures have been eased, please get in touch: Marjorie@ced.co.uk.  

 

If you see these buttons in our newsletters, it means a file or script relating to 

the section is available to download: 

Latest versions of Spike2 and Signal 

Spike2 Released  Signal Released 

Version 10.04 03/2020  Version 7.05a 02/2020 
Version 9.10 02/2020  Version 6.05b 10/2019 
Version 8.19a 11/2019  Version 5.12a 02/2018 
Demo 03/2020  Demo 02/2020 
     

Back to contents 

Future meetings and events 

Europhysiology 2020 
Berlin, 
Germany 
September 11th – 13th 2020 

Neuroscience 2020 
Washington, DC, 
USA 
October 24th – 28th 2020 

ISAC XXI 2020 
Lisbon, 
Portugal 
November 3rd – 6th 2020 

Back to contents 

Script Spotlight 

Our software team has developed a script which converts motion in a video recorded with the Spike2 multimedia 

facility (.avi or.mp4) into an "Activity" trace in the corresponding time view. The script does not attempt to track 

moving objects, but instead converts the video to grey-scale and interprets differences in intensity between frames, 

pixel by pixel, in a user-defined area of the image (for example as due to motion rather than flashing lights). The 

result is qualitative, with the output units as 'mean intensity change per pixel between video frames'. The maximum 

activity value is 100, where all pixels are undergoing maximum possible grey-scale change between frames; in 

practice activity values will be lower.  

This approach is not sensitive to colour. Therefore, movement in areas of high contrast and many contours will 

register disproportionately and changes in light intensity will register as movement. Despite these caveats, the script 

will be useful for applications such as sleep scoring of rats or mice moving around in relatively homogeneous 

environments. For example, an animal, or up to 6 animals in adjacent cages, can be videoed from above. The video is 

then processed offline with this script to create an activity trace for each animal. The area of the video frame to 

analyse for each animal may be specified by clicking and dragging a rectangle over the relevant area of the video 

frame to select it. The activity traces for all the animals can then be generated simultaneously. 

The activity traces could then be used as part of the input to a semi-automated sleep staging script (e.g. OSD4.s2s) 

based on a combination of activity level and banded power spectra. Other applications might include assessing time 

spent by a single animal in different parts of a behavioural arena. We appreciate any feedback for this script, and 

hope it is of benefit to you. 

Download the script from our website. 

 Back to contents 

 

 

 

 

 

 

http://ced.co.uk/upgrades/spike2upgrade/10
http://ced.co.uk/upgrades/spike2upgrade/9
http://ced.co.uk/upgrades/spike2upgrade/8
http://ced.co.uk/upgrades/spike2demo
http://ced.co.uk/upgrades/signalupgrade/7
http://ced.co.uk/upgrades/signalupgrade/6
http://ced.co.uk/upgrades/signalupgrade/5
http://ced.co.uk/upgrades/signaldemo
https://www.europhysiology2020.de/
https://www.sfn.org/meetings/neuroscience-2020
https://arterialchemoreceptors.org/
http://ced.co.uk/downloads/scriptspkanal

Spike2 stores memory channels in system memory. They are used for easily modifying data which cannot be achieved 

in disk channels, as these are optimized for recording large quantities of unchanging data.  Memory channels are 

widely used within scripts to create new channels of copied data or even entirely new script generated data. There is 

also the option to manually enter data into memory channels. 

Memory channels can be of any channel type and are displayed in the same way as other channels, i.e. a memory 

based level channel will appear the same as a disk based level channel. When you have finished modifying the data in 

the memory buffer it can be written to the data file as a disk channel or discarded; if you do not write memory 

channels to disk they are lost when the data file is closed.  

To create a new memory buffer, navigate to the analysis menu and select Memory Buffer -> Create New buffer. 

You’re presented with a simple dialog to choose the buffer type using the drop-down menu. Click OK to create an 

empty channel or click Import to copy data from a source channel. 

If you choose to import from a Waveform or RealWave channel when creating an 

event or marker buffer you are presented with the option to choose the mode of 

import. The modes available are Peaks, Troughs, Peaks and Troughs, Data rising 

through level, Data falling through level, and Rise and fall through level. The 

events or markers are created based on the Mode chosen, with the Size of the 

peak/trough or Level being supplied in Volts. The minimum interval between 

events in seconds also needs to be supplied, and the range of data to process can 

be specified using the start time and end time fields. For example, if one needed 

to import Peaks at least 100 ms apart which exceeded 1V in amplitude as an 

event I would choose the settings on the right:  

Creating a direct copy in memory of a disk channel is simpler. Navigate to the Analysis menu and select Memory 

buffer -> Create Channel Copy. Alternatively, right click the channel you wish to copy in the data and select Copy as 

memory channel. Specify the start and end time for the data to copy in the new dialog and click OK. If your source 

channel has been modified you will be presented with a warning. 

To manually add a new event item to an existing memory channel, navigate to the analysis menu and select Memory 

buffer -> Add items, or right click the memory channel and select Add Items from there. Choose the time of the data 

to add, the marker code if adding a marker, the value of the wave if adding a waveform, and click OK. Manually 

deleting an item from the memory buffer follows the same format as adding an item; from the Delete Items dialog 

you can choose to delete items around a time point, between a time range, or delete all items. 

Memory channels are useful for holding your data to manipulate. However, there is a limit to the quantity of data 

that can be held in memory and accessed quickly. 32-bit applications are limited to 4GB of memory. 64-bit 

applications can address much more. There is limited physical memory in your PC, maybe 8 or 16 GB. Once this is 

exhausted the operating system 'swaps out' what it thinks is currently unused physical memory to disk to make more 

room. This means that if you generate huge quantities of data in memory channels, all operations get slow as the 

system spends its time saving and restoring memory from disk. 

Back to contents 

 

 

 

What is a memory channel for, and how do I use 

them? 

Scripts: Spike2 

Following on from the introduction to memory channels above, we have combined many of the elements we have 

discussed in the past three newsletter issues to create a script which generates events at the point which waveform 

data crosses a threshold. This script creates a memory channel which uses the import channel function to generate 

these events. An idle function continually updates the memory buffer each time the threshold or time range is 

altered. 

Upon running the script, use the Open button in the script generated toolbar to load your data. Use the Help button 

to display guidance. We have used the Demo.smr file for our example here; this data file is stored in your user data 

folder, usually located in C:\Users\Username\Documents\Spike2\Data.  

 

After opening your file, you need to choose the channel you wish to generate events from by clicking the channel 

number in the Y axis (the script will show these for you if you have hidden them previously). You then need to add 

the threshold and optionally a time range before using the Add events button to create the memory channel.  

Much of the script is concerned with correctly selecting and positioning views and error checking. The main script 

element of interest is MemImport(chan%,inCh%,start,end{,mode%,time,level{,code%}}); this is 

contained in the Events%() toolbar function where start, end, and level are specified by the cursors on screen. The 

Mode is toggled using the associated button in the toolbar, and the minimum Time between events can also be 

specified via its button. The idle function continually checks for the current positions of the placed horizontal cursor 

(level) and the difference between the two vertical cursors (start and end) and compares them to their last known 

values. If these values differ the script runs the Update%() function which uses MemDeleteItem(chan% 

{,index% {,num%}}) to remove all items from the memory channel, and imports again with MemImport() using 

the new positions of the threshold and time gates. 

Download the script and try it out for yourself. We hope you find it useful. 

Back to contents 

 


'Events from Threshold Crossing|Script for generating events from waveform data crossing a threshold
'===========================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, TECHNICAL CENTRE, 139 CAMBRIDGE ROAD, MILTON, CAMBRIDGE CB24 6AX, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design.
'Author:	LS.
'v10 last modified :   20/04/2020
'SOFTWARE: This script was developed using Spike2 v10.0. It has been tested to work with Spike2 v8, v9 and v10

'HEALTH WARNING:
'The script is a -work in progress- and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'-----------------------------------------------------------------------------------------------------------------------------------

var chlst%[10]; 'array to hold channels
var vh1%;   'data view
var ch%;    'chosen channel
var thresh1%; 'threshold
var mode%:=2;   'import mode
var modeflag%:=0;   'debug flag
var RangeFlag%, eventflag%; 'debug flag
var low%, high%;    'vertical cursors
var mem%;   'memory channel
var time:=0.01, minimum:=10;    'time between events
var ThreshPosNew, ThreshPosOld; 'current and previous threshold position
var RangeOld, RangeNew; 'current and previous time gate positions
var msg%:=1; 'help flag, if 1 then hide messages
var Lh1%:=LogHandle();  'log view
var txt$;   'help text
var type%;  'type of help
var title$;

view(App(3)).WindowVisible(0); 'hide running script
ToolbarText("Placehold");
DoToolbar();   
Halt;

Func DoToolbar()    
ToolbarClear();     'Remove any old buttons
ToolbarSet(0, "", Idle%);   'Idle routine
ToolbarSet(1, "&Quit||Quit the script", Quit%);   'Link to function
ToolbarSet(2, "Open||Open a file", Open%);   'Link to function
ToolbarSet(3, "Threshold||Use this button to set the threshold", Threshold%);   'Link to function
ToolbarSet(4, "Set Range||Use this button to set a time range", Range%);   'Link to function
ToolbarSet(5, "Add events||Use this button to create a new channel holding events", Events%);   'Link to function
'ToolbarSet(6, "Release", Release%);   'Link to function
ToolbarSet(7, "Mode: Rising||Use this button to change the mode of threshold crossing", RiseFall%); 'Link to function
ToolbarSet(8, "Time: 10(ms)||Use this button to set the minimum time between events", Minimum%);    'Link to function
ToolbarSet(10, "Help||Display help text", log%);    'Link to function
ToolbarEnable(-1, 0);
ToolbarEnable(1, 1);
ToolbarEnable(2, 1);
ToolbarEnable(10, 1);
help%(1);
return Toolbar("Use the toolbar to generate events from data crossing your defined threshold", 1023);
end;

Func Idle%()   'Button 0 routine
If eventflag%=1 then
    viewlost%();
    if high%>0 or low%>0 then 
        if cursor(high%)<cursor(low%) then
            CursorRenumber();
        endif;
        RangeNew:=Cursor(high%)-Cursor(low%);
    endif;
    ThreshPosNew:=HCursor(thresh1%);
    If ThreshPosOld <> ThreshPosNew or RangeOld <> RangeNew then
        Update%();
    endif;
endif;
return 1;  'This leaves toolbar active
end;

Func Quit%()   'Button 1 routine
'CursorDelete(low%);
'CursorDelete(high%);
'HCursorDelete(thresh1%);
view(App(3)).WindowVisible(1);
halt;
return 0;  'This stops the toolbar
end;

Func Open%()   'Button 2 routine
var err%;
help%(2);
'You now need to select the desired waveform or realwave channel\nto process. To do so, click the channel number in the Y axis\nso it appears highlighted and then click OK
vh1%:=FileOpen("",0,1+2+8);
if vh1%<0 then return 1; endif;
if msg%=1 then
    view(vh1%).window(0, 0, 100, 100);
else
    view(vh1%).window(0, 20, 100, 100);
endif;
chlst%[0]:=0;
FrontView(vh1%);
title$:=view(vh1%).WindowTitle$();
ChanNumbers(1);
While chlst%[0]<>1 do
    'flash();
    err%:=Interact("Highlight the channel of interest by clicking on the channel number in the Y axis, then click OK", 1023, 0, "OK", "Quit");
    if err%=2 then
        return 1;
    endif;
    View(vh1%);
    ChanList(chlst%[],1+512+65536); 'list of selected waveforms/RealWaves
    docase
    case chlst%[0]=0 then Message("Error|No waveform or ReaLwave channel selected\nPlease retry");  
    case chlst%[0]>1 then Message("Error|More than one waveform or ReaLwave channel selected\nPlease retry"); 
    case chlst%[0]=1 then
        ch%:=chlst%[1];     'channel to process
    endcase;
wend

ToolbarEnable(3, 1);
ToolbarEnable(7, 1);
ToolbarEnable(8, 1);
help%(3);
return 1;  'This leaves toolbar active
end;

proc flash();
var i%;
for i%:= 1 to 20 do
chanselect(1, i%);
yield(0.1);
chanselect(1, 0);
yield(0.1);
next;
return;
end;

Func Threshold%()   'Button 3 routine
viewlost%();
help%(4);
View(vh1%);
if thresh1%<>0 then
    HCursorVisible(thresh1%, 1);
    HCursor(thresh1%, 0, ch%);
else
    thresh1%:=HCursorNew(ch%);
        if thresh1%=0 then
            Message("All horizontal cursors are in use, deleting one to repurpose as the threshold");
            HCursorDelete(1);
            thresh1%:=HCursorNew(ch%);
    endif;
endif
HCursorLabel(4, thresh1%, "Threshold - DRAG ME");
ToolbarEnable(4, 1);
ToolbarEnable(5, 1);
return 1;  'This leaves toolbar active
end;

Func Range%()   'Button 4 routine
var Q%;
viewlost%();
help%(5);
view(vh1%);
if RangeFlag%=1 then
    Q%:=Query("Do you wish to remove range gates?");
    if Q%=1 then
        CursorDelete(low%);
        CursorDelete(high%);
        low%:=0;
        high%:=0;
        RangeFlag%:=0;
        RangeNew:=0;
        RangeOld:=0;
        ToolbarSet(4, "Set Range||Use this to set a time range", Range%);
        help%(4);
        return 1;
    else
        return 1;
    endif;
endif;
'Drag these time gates to adjust the range for the generated\nevents.\n\nIf you wish to remove the gates entirely click the Remove Range\nbutton.
low%:=CursorNew(XHigh()/3);
CursorLabel(4, low%, "Lower gate - DRAG ME");
high%:=CursorNew(XHigh()*(0.66));
CursorLabel(4, high%, "Upper gate - DRAG ME");
if high%=0 or low%=0 then
    Message("All cursors are in use, deleting two to repurpose as the range gates");
    CursorDelete(low%);
    CursorDelete(high%);
    CursorDelete(1); 
    CursorDelete(2);
    low%:=CursorNew(XHigh()/3);
    CursorLabel(4, low%, "Lower gate - DRAG ME");
    high%:=CursorNew(XHigh()*(0.66));
    CursorLabel(4, high%, "Upper gate - DRAG ME");
endif;
RangeFlag%:=1;
ToolbarSet(4, "Remove Range||Remove the range and process entire channel", Range%);
return 1;  'This leaves toolbar active
end;

Func Events%()
help%(6);
viewlost%();
view(vh1%);
mem%:=Memchan(2);
ChanShow(mem%);
if RangeFlag%=1 then
    MemImport(mem%, ch%, Cursor(low%), Cursor(high%), mode%, time, Hcursor(thresh1%));
else
    MemImport(mem%, ch%, 0, maxtime(), mode%, time, hcursor(thresh1%));
endif;
ToolbarEnable(6, 1);
ThreshPosOld:=HCursor(thresh1%);
ThreshPosNew:=HCursor(thresh1%);
RangeNew:=Cursor(high%)-Cursor(low%);
RangeOld:=Cursor(high%)-Cursor(low%);
eventflag%:=1;
return 1;
end;

Func Release%()   'Button 5 routine
'Your code in here...
return 1;  'This leaves toolbar active
end;

Func RiseFall%();
If modeflag%=0 then
    mode%:=3;
    modeflag%:=1;
    ToolbarSet(7, "Mode: Falling||Use this button to change the mode of threshold crossing", RiseFall%);
else
    mode%:=2;
    modeflag%:=0;
    ToolbarSet(7, "Mode: Rising||Use this button to change the mode of threshold crossing", RiseFall%);
endif;
if eventflag%=1 then
    Update%();
endif;
return 1;
end;

Func Minimum%();
var ok%, dummy%;
var min$;
DlgCreate("Minimum time");
DlgLabel(1, "Enter minimum time between detected events");
DlgReal(2,"(ms)", 0, 1000000);
ok%:=DlgShow(dummy%, minimum);
if ok%=1 then
    time:=minimum/1000;
    min$:=Print$("%.3f(ms)||Use this button to set the minimum time between events", minimum);
    Toolbarset(8, min$, Minimum%);
endif
if eventflag%=1 then
    update%();
endif;
return 1;
end;

Func Update%();
view(vh1%);
MemDeleteItem(mem%, -1);
if RangeFlag%=1 then
    MemImport(mem%, ch%, Cursor(low%), Cursor(high%), mode%, time, Hcursor(thresh1%));
    RangeOld:=Cursor(high%)-Cursor(low%);
else
    MemImport(mem%, ch%, 0, maxtime(), mode%, time, hcursor(thresh1%));
endif;
ThreshPosOld:=HCursor(thresh1%);
return 1;
end;

func log%();
if msg%=1 then
    View(Lh1%);
    WindowVisible(1);
    Window(0, 0);
    WindowSize(100, 20);
    ToolbarSet(10, "Hide help||Hide the help text", log%);
    msg%:=0;
    If vh1% <> 0 then
        View(vh1%).Window(0,20, 100, 100);
    endif
else
    view(Lh1%).WindowVisible(0);
    view(vh1%).window(0, 0, 100, 100);
    ToolbarSet(10, "Help||Display help text", log%);
    msg%:=1;
endif    
'help%(1);
'help%(2); 'etc.
view(vh1%);
return 1;
end;

Func help%(type%);
view(Lh1%);
EditSelectAll();
EditClear();
docase
case type% = 1 then
    txt$:="This script will help you generate events from threshold crossings\n\nFirstly, use the Open button to open a data file.";
case type% = 2 then
    txt$:="You now need to select the desired waveform or realwave channel to process. Click the channel number in the Y axis so it appears highlighted and then click OK";
case type% = 3 then
    txt$:="Threshold: Use this button to set a new threshold.";
    Printlog(txt$);
    txt$:="\n\nMode: Use this button to change the type of threshold crossing.";
    Printlog(txt$);
    txt$:="\n\nRising = events are captured for data rising through the threshold.\nFalling = events are captured for data falling through the threshold.";
    PrintLog(txt$);
    txt$:="\n\nTime: You can also specify the minimum time between events, this is preset to 10ms meaning if two threshold crossings occur within 10ms of each other only the first is captured.";
case type% = 4 then
    txt$:="Drag the horizontal cursor to the level desired for the threshold.";
    PrintLog(txt$);
    txt$:="\n\nRange: Use this button to set a time range.\n\nAdd events: Use this button to create a new channel containing the events from the current threshold.";
case type% = 5 then
    txt$:="Drag these time gates to adjust the range for the generated\nevents.\n\nIf you wish to remove the gates entirely click the Remove Range button."
case type% = 6 then
    txt$:="Now you have created the events channel, you can still adjust the threshold and/or time range by dragging the cursors. The events will update automatically.";
    PrintLog(txt$);
    txt$:="\n\nThis channel can be saved to disk to store the events by right-clicking the channel and navigating to Write all memory buffer.";
    PrintLog(txt$);
    txt$:="\n\nYou can also create another memory channel by clicking the Add events button, and then move the cursors to update the time range and threshold.";
endcase;
Printlog(txt$);
return 1;
end;

Func viewlost%();
var Found%;
Found%:=ViewFind(title$);
if Found%=0 then
    Message("Cannot find data view, resetting script");
    vh1%:=0;
    eventflag%:=0;
    RangeFlag%:=0;
    modeflag%:=0;
    thresh1%:=0;
    DoToolbar();
endif
return 1;
end;



EventsFromThresholdCrossing.s2s
To download, right click and select Save embedded file to disk.

The automation tab of the sampling configuration provides this functionality. The tab controls three limits that will 

cause sampling to stop automatically: Number of Frames, Sampling duration, and File size. These limits only cause the 

sampling to stop, not finish; this means you have the choice of flipping through the frames and deciding whether to 

continue sampling or finish and save the data file. Clicking the More button in the sampling controls toolbar once 

sampling reaches a pre-set limit allows another limits-worth of sampling to occur before stopping again. For example, 

if the frame limit were set to 25, clicking More would cause sampling to continue until the data file had 50 frames. 

Clicking More again would give 75 frames and so on. 

Each option has a check box to enable the limit, plus a field for entry of the 

limit value. If the check box for a limit is clear, or if the corresponding limit 

value is set to zero, then that limit is ignored. 

Say for example we wished to obtain exactly 20 frames of evoked response 

data; we enable the Number of Frames limit by checking the box and enter 

20 into the dialog field. We create an evoked response experiment using 

multiple frame states, of which 10 of the 20 frames will have a single pulse 

delivered to a stimulator and the remaining 10 will have a paired pulse 

delivered. Upon reaching the limit, we check through the frames of data to 

see if the appropriate evoked response was achieved. In doing so we find 

anomalies on a few frames that we make note of and now wish to repeat.  

We first uncheck the Write to disk box and select Pause at end, swap to using Manual state control for the 

experiment and select the Idle state to prevent firing the stimulator. By clicking More, Signal is now ready to begin 

capturing more frames manually. We select the desired state in the state bar and click Continue to trigger another 

sweep. If the frame is good, we can choose to Accept this new frame to write it to disk. You can capture as many 

more frames as the original limit, so 

in this case another 20 to give a total 

of 40 frames. However, we can Stop 

early once we have gotten the 

correct frames and Finish once we 

are happy. 

The sampling configuration attached to this newsletter lets you try this out for yourself. 

Another automation tool within Signal is the Artefact rejection dialog. This is also 

accessed in the Automation tab of the sampling configuration. When enabled, 

Signal automatically examines newly sampled data. You have the choice to set 

Signal to either reject or tag new frames for when data has reached the ADC limits. 

Coupling automatic artefact rejection with your sampling limits is a useful aid in 

capturing the exact data you require. 

 

Back to contents 

 

 

 

 

How can I automatically end sampling in Signal? 


 
	 
		
		 
			 
				 
				 
			
		
		 
		 
		 
			 
				 
				 
				 
				 
				 
				 
				 
				 
			
		
		 
		 
		 
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
			
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
					 
				
				 
					 
				
			
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
					 
					 
				
				 
					 
				
			
		
		 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
		
		 
		 
			 
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
			
		
		 
		 
		 
		 
		 
	
	 
		 
			 
			 
			 
			 
			 
				 
			
			 
				 
				 
			
			 
			 
			 
		
	
	 
		 
	


NumOfFramesLimitExample.sgcx
To download, right click and select Save embedded file to disk.

Scripts: Signal 

In our last newsletter, we discussed setting up protocols for controlling stimulators from Signal. It can be sometimes 

be useful to import a previously generated protocol through other means into Signal. Rather than copying the 

protocol verbatim, this script enables the user to generate a protocol for a Magstim BiStim stimulator as a text file 

and import the settings as needed. The script will condense the protocol into the minimum number of states required 

to play out all steps of the protocol. The correct timings of the stimulator firing and the intensity are kept for each 

step. 

The format required for the text file protocol is as follows: 

POWER A%   POWER B%    PULSE INTERVAL(ms)  SWEEP INTERVAL(s) 

The power % can be either a value of Maximum Stimulator Output (MSO) from 0-100%, or as a percentage of motor 

threshold (MT) %. If you specify the latter, you must also provide the MT % value; you are given the choice to specify 

either working in MSO or MT when importing the protocol.  

In the text file, each element MUST be separated by a tab for the script to correctly read each setting. A text file using 

POWER% as a percentage of MT, providing three instructions, would look something like the example below. We 

have also included a few example text files to download with the script to try out for yourself: 

80 120 15 7 

80 120 3 6 

80 0 0 0 

To use the script, ensure your BiStim’s Trigger Out is connected to the Trigger port of your 1401. You must also set 

the BiStim to Independent Triggering mode (also known as IBT mode), connect Digital Output 0 to the master 

stimulator (Power A), and connect Digital Output 1 to the secondary stimulator (Power B). When running the script 

you need to first use Load Protocol. After selecting your protocol, the script calculates how many states are required 

and the order of the states needed to play out each instruction. The Sample button is then enabled, when clicked the 

script asks you to provide a trigger before sampling will begin. You may progress the script to this point before setting 

up the stimulator, but your stimulator must be connected to the 1401 as described above before continuing. When 

ready, manually fire a pulse from the BiStim to provide the trigger to the 1401, which then prompts the script to 

check all is OK before finishing the set up. Sampling then proceeds using the loaded protocol.  

Whilst this script only currently functions with a Magstim BiStim, it is possible to modify the script to work with other 

Signal supported devices. If this is something you are interested in, please get in touch: info@ced.co.uk. 

Back to contents 

Did you know…? 

Scale bars are available to use within Spike2 and Signal to help create useful figures for your reports. To enable scale 

bars, right click the data and click Show/Hide channels for Spike2 or Customise display for Signal. Here you are given 

options for customising the Y and X axis. Ticking the Scale bar option under either axis list will enable it. It’s best to 

play around with your other draw settings as well to better portray your data, but if you get lost you can always 

revert to the default settings by navigating to the View menu and selecting Standard display. 

Back to contents 

Recent Questions 

How do I create a long train of several thousand pulses? 

There are several ways one could create a large train of pulses within Spike2 and Signal. The graphical pulse editor 

may be used to create a pulse train of up to 1000 pulses, the repeat field would then be used to repeat the train 

however number of times is needed to create the long train. However, this method less efficient as each pulse is 

stored as an instruction to be played out. It is preferred to specify a smaller number of pulses in one section and 

repeat the required number of times. It is even more efficient to delve into the sequencer language to create a single 

'BiStim Protocol control|Script for importing text protocol for BiStim control
'===========================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, TECHNICAL CENTRE, 139 CAMBRIDGE ROAD, MILTON, CAMBRIDGE CB24 6AX, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design.
'Author:	SG.
'Last modified :   20/04/2020,  LS.
'SOFTWARE: This script was developed using Signal v5. It has been modified and tested to work with Signal v7

'HEALTH WARNING:
'The script is a -work in progress- and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'-----------------------------------------------------------------------------------------------------------------------------------

'OVERVIEW
'This script enables the user to generate a protocol for the TMS device as a text file and import the settings as needed. 
'The script will condense the protocol into the minimum number of states required to play out all steps of the protocol. 
'The correct timings of the stimulators firing and the intensity are kept for each step. The format for the protocol in a 
'text file is as follows:

'POWER A%   POWER B%    PULSE INTERVAL(ms)  SWEEP INTERVAL(s)

'Note that the POWER% can be either a value of Maximum stimulator output (MSO) from 0-100%, or as a value of motor threshold (MT) %
'Each element MUST be separated by a tab for the script to correctly read each setting. When importing the protocol, you have the 
'choice to specify either working in MSO or MT
'An example text file using POWER% as MT would look something like:

'80 120 15  7
'80	120 3   6
'80 0   0   5

'-----------------------------------------------------------------------------------------------------------------------------------

'INSTRUCTIONS
'It is important the BiStim is connected to the 
'When running the script it loads a toolbar:
' CONTINUE || PAUSE ON/OFF || SAMPLE || LOAD PROTOCOL || QUIT

'QUIT: This button quits the script
'
'LOAD PROTOCOL: This button will first present a dialog allowing you to specify the %MT and the number of repeats required for 
'               the protocol. If you wish to work in MSO% then tick the associated box which will grey out the MT field. Upon
'               clicking OK you are presented with the File Open dialog, which allows you to locate and select the protocol.

'SAMPLE: This button is enabled once you have loaded a protocol and is used to start sampling. You will first be asked to supply
'        a trigger to check the BiStim is connected. Ensure the TRIGGER OUT of the BiStim is connected to the TRIGGER port of the
'        1401, and fire the BiStim manually. You are then asked to select the COM port for the BiStim. If there is more than one 
'        COM port available and you do not know which COM port to use, you will need to open your Device Manager and locate the 
'        BiStim device under PORTS. When you have selected the correct COM port the script will attempt to create a sampling 
'        configuration to check all is ready. You are then presented with a 10 second countdown until the start of sampling. After 
'        the countdown, the loaded protocol will begin running.

'PAUSE ON/OFF: Use this button to toggle the PAUSE AT END OF SWEEP function. If on, sampling will pause at the end of each sampling
'              sweep. You would then use the CONTINUE button to continue to the next sweep. However bear in mind that using this 
'              function essentially renders the SWEEP INTERVAL element of the protocol void, unless CONTINUE is pressed before the
'              SWEEP INTERVAL time has finished.

Var data%;           'Data file handle
Var Basic%:=1;
Var Out%[5][0];
Var Order%[1];
Var Times[1];
Var States%[5][1];
Var NumStates%;
Var threshold%:=32;
Var GlobalIdx%:=0;
Var LastFrCt%:=0;
Var go%:=0;
Var rep%:=1;        'Number of repeats'
Var Pause%;         'Pause on/off flag
Var MSO%;
var gFloat%[20];    'App() view handles
var Aux%;
var port%;	        'Serial port number
Var Ports$[20];     'Array to hold availble serial ports
Var ct%;

ToolbarText("");

HideApps();
CheckSerialPorts%();
DoToolbar();
Halt;

Func DoToolbar()    'Set your own name...
ToolbarClear();     'Remove any old buttons
ToolbarSet(0, "", Idle%);   'Idle routine
ToolbarSet(1, "Quit", Quit%);    'This button returns its number
ToolbarSet(2, "Load Protocol", LoadTxt%);   'Link to function
ToolbarSet(3, "Sample", Sample%);   'Link to function
Toolbarset(5, "Pause OFF", PauseState%);
ToolbarSet(6, "Continue", Continue%);
ToolbarEnable(3, 0);
ToolbarEnable(5, 0);
ToolbarEnable(6, 0);
return Toolbar("Select...", 1023);
end;

Func Sample%()   'Button 3 routine
Var err%;

'If Basic% =  1 then
Err%:=SampleBasic%();

If err%<0 then
    DoToolbar();
endif

    GlobalIdx%:=0;

    SetConfig%();
    SetupStates%();    
    SampleProtocol%();


return 1;  'This leaves toolbar active
end;

Func UpdateSwp%(ct%);
Var err;
Var interval;

GlobalIdx%:=GlobalIdx%+1;
If GlobalIdx% = Len(Order%[]) and rep% > 1 then
'    Samplestop(1);
    GlobalIdx%:=0;
endif    
    'Printlog("\nUpdateSwp% end %d\n", GlobalIdx%);
If GlobalIdx% < Len(Times[]) then
    Interval := Times[GlobalIdx%];
'If interval = 0 then interval := 1; endif 'Check if the last interval is zero
    SampleFixedInt(Order%[GlobalIdx%], interval); 'N seconds interval between states
    Err:=SampleFixedInt(Order%[GlobalIdx%]); 'N Seconds interval between states
    SampleState(Order%[GlobalIdx%]);
endif

If ct% < Len(Order%[])*rep% then
        ToolbarText(Print$("Next State %d. Current Stim #%d of %d", Order%[GlobalIdx%], ct%, (Len(order%[])*Rep%))); 'This is running 1 behind
    else
        ToolbarText(Print$("Finished. Current Stim #%d of %d", ct%, (Len(order%[])*Rep%))); 'This is running 1 behind
    endif

Return 1
end

Func PauseState%();

If Pause% = 0 then
    Pause%:=1;
    SamplePause(1);
    Toolbarset(5, "Pause ON", PauseState%);
    ToolbarEnable(6, 1);
else
    Pause%:=0;
    SamplePause(0);
    Toolbarset(5, "Pause OFF", PauseState%);
    ToolbarEnable(6, 0);
endif

Return 1
end

Func Continue%()
if data%>0 then
    if go% = 1 then
        Pause%:=0;
        SampleSweep();
        Idle%();
        Pause%:=1;
    endif
endif

return 1;
end;

Func Idle%()   'Func operates when script is not busy

Data%:=SampleHandle(0); 'Get the handle of the samplnig document

If data% > 0 then 'If a file exists
    View(data%);

    
    If Pause% = 0 and go% = 1 then        
        'If basic% = 0 then
            
        'If go% = 1 then
 
            ct%:=Framecount();
            
            If ct%>LastFrCt% then
                'Check end
                
                If ct% < Len(Order%[])*rep% then
                    UpdateSwp%(ct%);
                    LastFrCt%:=ct%;
                else
                    go%:=0;
                    SampleStop();
                    UpdateSwp%(ct%);
                    Message("Finished...");
                    Data%:=0;
                    Basic%:=1;
                Endif
                
            endif     
         
    endif
endif    

return 1;  'This leaves toolbar active
end;

Func DeleteFile%();

View(data%);
SampleAbort(1);
'Fileclose(0,-1);
Data%:=0;
Basic%:=0;
Sample%();

Return 0
end


Func LoadTxt%();
Var ok%;

DlgCreate("Setup");  'Start new dialog
DlgAllow(1023, 0, Change%);
DlgInteger(1,"Subjects Threshold level",5,100);
DlgInteger(2,"Number of cycles (repeats)",1,10);
DlgCheck(3, "% Maximum stimulator output? Otherwise % RMT");
ok% := DlgShow(threshold%,Rep%, MSO%);

If ok% = 0 then
    Dotoolbar();
endif

ReadProtocol%();
ToolbarEnable(3, 1);

Return 1
end

Func Change%(item%);

If item% =  3 or item% = 0 then
    
    if Dlgvalue(3) > 0 then
        Dlgenable(0, 1);
    else
        Dlgenable(1, 1);
    endif    
endif      

Return 1
end



Func SetupStates%();
Var State%;
Var interval;
Var data%[4];

'Arrsort(States%[0][],0,States%[1][],States%[2][],States%[3][]);

For State%:= 1 to NumStates% do
    Arrconst(data%[], States%[][State%-1]);    
    Interval:=Data%[3];  
    
    If interval = 0 then interval := 1; endif    'Traps zero interval time.
    
    SetOutputPulses%(State%, Data%[]);
    LoadMagstim%(State%, data%[1], data%[2], Interval);
Next    

Return 1
end

Func LoadMagstim%(State%, Pwr1%, Pwr2%, Interval);
Var ReadBkPwr1%, ReadBkPwr2%;
Var PwrPulse1%, PwrPulse2%;
Var Type%, Model%;

Type%:=SampleAuxStateParX(Aux%,0); 'check stimulator is a magstim
Model%:=SampleAuxStateParX(Aux%,1); 'check its a bistim

If Type% <> 1 and Model% <> 1 then
    Message("No Magstim enabled. Halting");
    halt;
endif

If Interval=0 then interval:=5; endif
'MotorThresh*Pwr1%/100;
SampleAuxStateValX(Aux%, State%, 0, 0); 'Not manual controlled

If MSO% = 1 then
    PwrPulse1%:=Pwr1%;
else
    SampleAuxStateParX(Aux%, 7, 1);
    SampleAuxStateParX(Aux%, 8, threshold%);
    PwrPulse1%:=Pwr1%;
endif    

SampleAuxStateValX(Aux%, State%, 1, PwrPulse1%); ''''Need threshold level...
ReadBkPwr1%:=SampleAuxStateValX(Aux%, State%, 1);

If MSO% = 1 then
    PwrPulse2%:=Pwr2%;
else
    SampleAuxStateParX(Aux%, 7, 1);
    SampleAuxStateParX(Aux%, 8, threshold%);
    PwrPulse2%:=Pwr2%;
endif  

SampleAuxStateValX(Aux%, State%, 2, PwrPulse2%);
ReadBkPwr2%:=SampleAuxStateValX(Aux%, State%, 2);
'SampleAuxStateValX(Aux%, State%, 3, Interval);

If ReadBkPwr1% < 5 then
    Message(Print$("Bad intensity level < 5 percent set. Power 1 set to %d. Should be %d...", ReadBkPwr1%, PwrPulse1%));
Endif

Return 1
end

Func SampleBasic%();
Var err%;
Var Info%;

SampleClear();
SampleRate(5000);'
SamplePorts(1);
SamplePoints(0.3*SampleRate());
SampleMode(0); 'basic mode
SampleStatesMode(2); 'dynamic output
SampleZeroOffset(0.1);
SampleStates(0); 'no extra states
SampleStatesOrder(0); 'Numeric output
Err%:=SampleAuxStateParX();
If Err%=10 then
    Message("Auxilliary device list full, please remove one to make room\nHalting script");
    halt; endif;
Aux%:=SampleAuxStateNew(1);
SampleAuxStateParX(Aux%, 1, 1); 'set bistim
SampleStatesIdle(0);
SampleWrite(1);
SampleBurst(1);
SampleTrigger(1);
if Pause%=0 then
    SamplePause(0);
else
    SamplePause(1);
endif;
Data%:=Filenew(0);

If data% < 0 then
    Message(Print$("Sampling setup failed.\n%s",Error$(Data%)));
    Return -1
endif

View(app(6)).WindowVisible(0);
View(app(7)).WindowVisible(0);
View(app(9)).WindowVisible(0);
View(data%);
XAxis(0); YAxis(0);
ChanNumbers(0);
Window(0,0,100,100);
Windowvisible(0);
Info%:=InfoNew("Waiting for pulse at 1401 Trigger input\nto continue", 8+16+32); 
FrontView(Info%);
Window(0,0,100,100);
ToolbarText("Waiting for trigger input...");
Samplestart();
View(data%);
While FrameCount() = 0 do Yield(0.1) wend;
Data%:=Sampleabort(1);

ToolbarText("Setting up sampling...");


Return 1
end

Func SampleProtocol%();
Var err;
Var StpWtc%;
Var Name$, Size, Flags;
Var curT%, lastS%:=-10;
Var err%;
Var path$;
Var col%:=0;
Var Red:=0.1;
Var Green:=0.9;

Path$:=View(app(3)).FileName$(1)+View(app(3)).FileName$(2);
Data%:=Filenew(0);

If data% < 0 then
    Message(Error$(Data%));
    Return 1
endif

ToolbarEnable(5, 1);

View(app(6)).WindowVisible(0); 'Hide the sample bar
View(app(7)).WindowVisible(0); 'Hide the sample control panel
'View(app(9)).WindowVisible(0); 'Hide the states control bar
View(app(9)).Windowvisible(1);
View(data%);
Window(0,0,100,100); 'Size the window
WindowVisible(1);
FrontView(Data%);
SampleFixedInt(Order%[GlobalIdx%], Times[GlobalIdx%]); 'N seconds interval between states
Err:=SampleFixedInt(Order%[GlobalIdx%]); 'N Seconds interval between states
SampleState(Order%[GlobalIdx%]);

LastFrCt%:=0;

StpWtc%:=InfoNew();
InfoSettings("   %t", 1+2+4+8+32, 10, 0, 0.1, 1);

Window(0,0,100,100);

ToolbarText("10 second delay. Please wait...");
Seconds(-10);
InfoReset();
While seconds() < 0 do Yield(0.1);
    ToolbarText(Print$("10 second delay. Please wait... %.2F", Seconds()*-1));
    CurT%:= Trunc(seconds());
    
    if CurT% > lastS% then
        ViewColourSet(0, Red, Green, 0);
        Red:=Red+0.1;
        Green:=Green-0.1;
        Col%:=Col%+1;
        lastS%:=CurT%;        
    endif  
      
wend;


FileClose(0,-1);

FrontView(data%);
ToolbarText(Print$("Next State %d. Current Stim #%d of %d", Order%[0],(Len(order%[])*Rep%)));
Samplestart();
Go%:=1;


Return 1
end

Func SetConfig%();
Var err%;

SampleClear();
SampleRate(5000);'
SamplePorts(2);
SamplePoints(0.3*SampleRate());
SampleMode(3);
SampleStatesMode(2);
SampleZeroOffset(0.1);
SampleStates(0);
SampleStates(NumStates%);
SampleStatesOrder(0); 'Numeric output
SampleStatesOptions(2); 'Writing but no cycling
SampleStatesRun(0); 'Manual mode - not cycling
SampleStatesRepeats(1);
Err%:=SampleAuxStateParX();
If Err%=10 then
    Message("Auxilliary device list full, please remove one to make room\nHalting script");
    halt; endif;
Aux%:=SampleAuxStateNew(1);
SampleAuxStateParX(Aux%, 1, 1); 'Set as bistim
SampleAuxStateParX(Aux%, 2, 8); 'set as independent triggers
SampleAuxStateParX(Aux%, 3, Port%); 'set port number
SampleStatesIdle(1);
SampleWrite(1);
SampleBurst(1);
SampleTrigger(1);
If Pause%=0 then
    SamplePause(0);
else
    SamplePause(1);
endif;
Return 1
end

Func CheckSerialPorts%();
Var i%;
var ctp%:=0;
Var err%;
Var list%:=1, ok%;
For i%:= 1 to Len(Ports$[]) do
    err%:=Serialopen(i%);
    If err% = 0 then
        Ports$[i%]:=Print$("%d", i%);
        ctp%+=1;
        SerialClose(i%); 
    endif
Next
Ports$[0]:="None";
Arrsort(Ports$[1:], 1);         'Sort in decending order
If ctp% > 0 then
    ArrSort(Ports$[1:ctp%], 0); 'Sort in ascending order
    Resize Ports$[Ctp%+1];
endif  

DlgCreate("Setup");                                     'Start new dialog
DlgList(1,"Serial port to use",Ports$[0:ctp%+1]);
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(list%);                                  'ok% is 0 if user cancels,  variables updated if not
If ok% = 0 or List% =  0 then
    Message("No Com port. Halting");
    Quit%();
endif
If list% > 0 then
    Port%:=Val(Ports$[List%]);
endif    
Return 1
end

Func SetOutputPulses%(State%, Data%[]);
'Var i%;
Var pulseNum%;
Var interval;
var err;

SampleOutMode(1);
SampleDacMask(0);
SampleDigOMask(3);
'for i%:= 1 to NumIndvStates% do
SampleOutLength(State%, 0.3);
SampleOutTrig(State%, 0.0);
SampleFixedInt(State%, 7); 'Seven seconds interval between states
SampleFixedVar(State%, 0);
pulseNum% := PulseAdd(State%, -1, 1, "First", SampleOutTrig(State%)+0.1, 0.001);
PulseDataSet(State%, -1, pulseNum%, 1);

'if Data%[2] > 0 then
err:=SampleoutTrig(State%);

if Data%[3] > 0 then
    Interval:=Data%[3];
else
    Interval:=5;
endif    

if data%[2] <> 0 then
pulseNum% := PulseAdd(State%, -1, 1, "Second", SampleOutTrig(State%)+0.1+(interval/1000), 0.001);
err:=PulseDataSet(State%, -1, pulseNum%, 2);
endif
'Next 

Return 1;
end

Func ReadProtocol%();
Var txt%;
Var err%;
Var arr%[4];
Var Sort%[4][0];
Var i%:=0;
Resize Out%[][0]; 'Add another row for the new data
Var States%[5][1];

Arrconst(States%[][],0);
Txt%:=Fileopen("", 8, 8);

If txt% < 0 then
    Message("File failed to open.");
    DoToolbar();
endif

NumStates%:=0;
ReadSetup("\t","\t","\t","\t","\t");

Repeat
    Err%:=Read(arr%[]);
    If err% > 0 then
        i%:=i%+1;
        SortValues%(arr%[],i%);
    endif    
Until Err% < 0;

Fileclose(txt%);

Return 1;
End

Func SortValues%(arr%[], i%);
Var j%:=0, k%;
Var check%[5];
Var Match%:=0;
Var num%;
Var got%:=0;
Var err%;
Var St%;
Var entry%;
Var int;

Resize Out%[][i%]; 'Add another row for the new data
Resize Order%[i%];
Resize Times[i%];  

entry%:=i%-1;

If MSO% = 1 then 'If using MSO which has the format intensity, sweep interval.
    Arr%[Len(arr%[])-1]:=Arr%[1]; 'Move the sweep interval to the correct position
    Arrconst(Arr%[1:1], 0); 'Delete the original value
endif    

Arrconst(Out%[1:4][i%-1], Arr%[]); 'Place the new values
Int:=Arr%[3]; 'holds interval

For j% := 0 to NumStates%-1 do; 'Step through the known states
    
    Match%:=0;
    for k%:= 1 to 3 do
        err%:=States%[k%][j%];
        
        If States%[k%][j%]=Arr%[k%-1] then
            Match%:=Match%+1;
            St%:=j%;
        else k%:=3;
        endif    
    next
    
    If match%=3 then
        j%:=NumStates%;
        Got%:=1;
    else
        Got%:=0;
    endif    
Next

If got%= 1 then
    Out%[0][Entry%]:=St%+1;
    Order%[Entry%]:=St%+1;
    Times[Entry%]:=Int;    
else
    NumStates%:=NumStates%+1;
    Resize States%[][NumStates%];
    Arrconst(States%[1:4][NumStates%-1], Arr%[]);
    States%[0][NumStates%-1]:=NumStates%;
    Out%[0][Entry%]:=NumStates%;
    Order%[Entry%]:=NumStates%;
    if int = 0 then int:=1; endif
    Times[Entry%]:=Int;  
endif

Return 1;
end

Proc HideApps();    'Hide unnecessary toolbars etc. while script is running.
var i%;

View(App(3)).WindowVisible(1);  'normalise script view before hiding (may reappear if maximised)
gFloat%[0] := App(-2);  'number of windows

for i% := 1 to gFloat%[0] do    'hide all windows and save state
	gFloat%[i%] := View(App(i%)).WindowVisible(0);
next;
return
end;

Proc RestoreApps(); 'Restore toolbars etc. to former glory
var i%;

for i%:=1 to gFloat%[0] do
	docase
	case i%=3 then
		View(App(3)).Windowvisible(0);  'keep script invisible
	else
		View(App(i%));  'restore other items to previous state
		WindowVisible(gFloat%[i%]);
	endcase;
next;
return
End

Func Quit%()   'Button 1 routine
RestoreApps();
Halt;
return 0;  'This stops the toolbar
end;

BiStim Protocol control v7.sgs
To download, right click and select Save embedded file to disk.

pulse which is repeated the desired number of times, as you can then use the minimum number of sequencer 

instructions. The sequencer is limited to 8191 instructions, although you are unlikely to reach this it is still preferable 

to use the least amount of instructions per section as it leaves more room for further instructions should your 

experiment evolve and require more later. 

In the sequencer language, a train of 5000 pulses of 1ms length and 1ms interval could be created as this: 

                SET    0.1,1,0         ;Get rate & scaling OK 

 

                VAR    V45,Loop=5000   ;Define variable for section loops 

 

0000            HALT    

0001 LA:    'A  DIGOUT [.......1]      ;Loopback point 

0002            DELAY  s(0.0009)-1 

0003            DIGOUT [.......0] 

0004            DELAY  s(0.0008)-1 

0005            DBNZ   Loop,LA         ;Repeat required times 

0006            HALT                   ;End of this sequence section 

 

Back to contents 

CED User forums 

Try the CED Forums bulletin board for software and hardware support. 

If you have any comments about the newsletter format and content, please get in touch: Marjorie@ced.co.uk. 

To adjust your subscription preferences, please visit our website: www.ced.co.uk/upgrades/subscribeenews. 

Contact us: 
 
In the UK: 
Technical Centre, 139 Cambridge Road, 
Milton, Cambridge, CB24 6AZ, UK 
Telephone: (01223) 420186 
Fax: (01223) 420488 

 
 
Email: info@ced.co.uk 
International Tel: [44] 1223 420186 
International Fax: [44] 1223 420488 
USA and Canada Toll Free: 1 800 345 7794 
Website: www.ced.co.uk 

 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

Copyright © 2020 Cambridge Electronic Design Ltd, All rights reserved. 

